

Alcune considerazioni sul concetto di "precisione" in computer grafica

In questo numero tratteremo un argomento tipico della Computer Grafica: la precisione. Utilizzeremo, per quanto riguarda la precisione nell'input, sia la tavoletta grafica di MC che il nuovissimo Digitizer DT1000 della Watanabe, che abbiamo potuto provare e che utilizzeremo anche nei prossimi numeri.

La precisione è un concetto che abbiamo tutti ben chiaro e con il quale dobbiamo fare i conti quando affrontiamo problemi in cui occorre eseguire delle misure o dei calcoli.

In computer grafica la "precisione" riguarda sia la fase di input dati, sia la fase di elaborazione, sia la fase di output.

E poiché, come dice un detto popolare, non c'è catena più forte del suo più debole anello, occorre equilibrare la precisione delle tre fasi, in quanto è inutile spingere una di esse quando un'altra fase ha una precisione bassa.

In sostanza è inutile utilizzare numeri con precisione a nove cifre quando tali numeri rappresentano ad esempio punti da visualizzare su un video con precisione di qualche centinaio di punti.

D'altro canto altre grandezze, che non siano coordinate di punti da visualizzare ma che siano elementi della elaborazione, è indispensabile che abbiano la massima precisione. Ad esempio mentre è sufficiente utilizzare numeri interi per definire le coordinate di un punto, non si possono certo usare numeri interi per definire, in radianti, gli angoli che intervengono in certi calcoli.

Per quanto riguarda la fase di input dei dati grafici, se i dati sono immessi via tastiera e inseriti direttamente nel programma possiamo scegliere la precisione che vogliamo. Se utilizziamo invece gli apposi-




Figura 1 - La tavoletta grafica di MCmicrocomputer. È stata presentata più volte nei numeri scorsi della nostra rivista.

ti strumenti di input (in genere il digitizer o le paddle) entra in gioco la loro precisione.

La tavoletta grafica di MCmicrocomputer, essendo collegabile all'Apple II, è stata realizzata in modo da avere la stessa precisione del suo monitor grafico. Ha quindi un piano di lavoro di 280 per 192 millimetri e la precisione di puntamento è di un millimetro, che tradotto sul video equivale al faticoso pixel.

L'anello iniziale della catena, digitizer, e l'anello finale, monitor, sono dunque perfettamente equilibrati.

Il Digitizer Watanabe Mitablet DT1000 ha ovviamente prestazioni migliori. Nella scheda tecnica qui a fianco ne specifichiamo le caratteristiche tecniche.

La tavoletta grafica di MCmicrocomputer

Prima di passare ai programmi che eseguono le prove comparate delle tre modalità di input (tastiera, tavoletta grafica di MC, Watanabe Mitablet DT1000), presentiamo, isolate dal contesto di un pro-

gramma, le routine di immissione dati della tavoletta di MC (listato fig. 2) e del DT 1000 (listato in fig. 3), utili per chi voglia sapere come lavora un digitizer.

La tavoletta di MC è stata più volte presentata e utilizzata nei numeri precedenti di MC.

Il primo programma richiede inizialmente il caricamento dei dati dal file "Paddle code" (riga 110) in cui sono memorizzati tra gli altri gli elementi ottenuti con il programma di calibrazione, e sono definite funzioni e costanti.

Il main program è la sola riga 160 che richiama via via la routine di immissione del singolo punto e visualizza i valori trovati.

La routine di immissione singolo punto è a riga 170 e comporta il richiamo a sua volta della sottoroutine di lettura continua dei valori inviati dalle paddle (ricordiamo che la Tavoletta si collega alla game-port dell'Apple II).

La condizione di riga 190 (che si verifica quando si preme il bottone della tavoletta) provoca l'interruzione della lettura continua e il calcolo dei valori X%, Y% (righe 210 - 230).

Ricordiamo infine che non essendo questo un programma grafico non vi sono routine di formattazione e i dati possono assumere valori negativi.

Il Miplot DT1000

Questo secondo programma comprende la routine di caricamento dei dati dal DT1000. È la routine di riga 210-270.

In riga 230 c'è un PRINT CHR\$(7), in quanto è bene che un BEEP avverta quando si può immettere un dato. Vedremo, in uno dei prossimi numeri, che per un lavoro di immissione accurato punto per punto è opportuno che il Digitizer lavori in POINT MODE, ovvero accetti il dato solo quando viene premuto un tasto sul tastierino. Altrimenti c'è lo STREAM MODE, in cui i dati vengono inviati con continuità, con una frequenza stabilita tramite una apposita istruzione di temporizzazione.

Il digitizer invia i dati sotto forma di caratteri ASCII, secondo il formato XXXX, YYYY, F cr. Questi caratteri sono caricati nella stringa K\$(di riga 250),

```

100 REM INIZIALIZZAZIONE
110 PRINT CHR$(4)"BLOAD PADDLE.CODE"
120 DEF FN PK(I) = PEEK (I) + 256 * PEEK (I + 1)
130 Z0 = FN PK(797):Z1 = FN PK(799)
140 V0 = FN PK(801):V1 = FN PK(803)
150 PY = 3.14159 / V1:PZ = 3.14159 / V0
160 REM MAIN PROGRAM
170 GOSUB 180: PRINT X%, Y%: GOTO 170
180 REM IMMISSIONE PUNTO T
190 GOSUB 230: IF PEEK (49251) > 127 THEN 190
200 A = (Z0 - P0) * PZ: B = (P1 - Z1) * PY - A
210 X% = 150 * (COS (B) - COS (A))
220 Y% = 150 * (SIN (B) + SIN (A)): RETURN
230 REM LETTURA CONTINUA VALORI PADDLE
240 POKE 779,100: CALL 768:P0 = FN PK(12)
250 FOR K = 1 TO 50: NEXT K
260 POKE 779,101: CALL 768:P1 = FN PK(12)
270 FOR K = 1 TO 50: NEXT K: RETURN
    
```

Figura 2 - Programma di input della tavoletta di MC. È un programma che produce semplicemente il PRINT sul video delle coordinate immesse via Tablet.

```

100 REM INIZIAL FATTORI DI SCALA
110 GOSUB 190: SX = 280 / 3800: SY = 160 / 2600
120 NP = NP + 1: GOSUB 210: HOME: VTAB (23)
130 PRINT "PUNTO N.": NP: TAB (14): X%: TAB (19): Y%: TAB (26): F%
140 X% = X% * SX: Y% = 159 - Y% * SY
150 IF F% = 1 THEN HPL0T X%, Y%: R% = X%: B% = Y%: GOTO 180
160 IF F% = 4 THEN HPL0T X%, Y%: HPL0T TO R%, B%: GOTO 180
170 HPL0T TO X%, Y%: IF F% < > 2 THEN END
180 GOTO 120
190 HOME: HGR: HCOLOR = 3
200 HPL0T 0, 0 TO 279, 0 TO 279, 159 TO 0, 159 TO 0, 0: RETURN
210 REM LETTURA DA DIGITIZER
220 REM WATANABE MITABLET DT1000
230 PRINT CHR$(7): REM AUTORIZZ. INPUT
240 K$ = "": IN# 2: REM SLOT
250 FOR I = 1 TO 12: GET J$: K$ = K$ + J$: NEXT I
260 X% = VAL ( MID$( K$, 1, 4) ): Y% = VAL ( MID$( K$, 6, 4) )
270 F% = VAL ( MID$( K$, 11, 1) ): IN# 0: RETURN
    
```

Figura 3 - Programma di input per la Mitablet DT1000. È un programma che presenta una uscita alfanumerica, nelle quattro righe in basso, e una uscita grafica sul monitor HGR dell'Apple II.




Figura 4 - Disegno usato per il test degli input. Per testare le tre forme di input abbiamo utilizzato un identico disegno, che serve per la taratura, per il calcolo dell'area F1, e per il calcolo degli angoli della figura F2.

che viene successivamente frazionata e tradotta in valori numerici.

L'istruzione IN#2 trasferisce la funzione di input dalla tastiera al digitizer che è inserito nello SLOT 2. La corrispondente istruzione di "SWITCH OFF" è l'IN#0. Questo programma, al contrario del precedente, lavora sul monitor grafico dell'Apple II. Per tradurre i dati dal formato Digitizer (che è di 380 per 260 millimetri) al formato video HGR (che è di 280 per 160 pixel) occorre trasformarli secondo i due fattori di scala SX (che è pari a 0.07368...) e SY (0.06153...).

Accettiamo cioè che pur di lavorare a pieno schermo le proporzioni della figura siano schiacciate lungo l'asse Y.

I dati immessi sono visualizzati in forma alfanumerica (riga 130) e visualizzati in forma di disegno, secondo una logica dipendente dal valore del flag F%.

F% = 1 inizio spezzata
F% = 2 continuazione spezzata

F% = 4 fine della spezzata e chiusura della figura con collegamento al punto di inizio spezzata.

F% = 8 fine della immissione.

In tale maniera ad esempio per disegnare un triangolo occorrerà immettere:

X1,Y1,1 primo punto e memorizz., coordinate in A%, B%

X2,Y2,2 secondo punto, collegato al primo

X3,Y3,4 terzo punto collegato sia al secondo che al primo.

Questo è un piccolo esempio di programma senza menu indirizzabile sulla superficie della tavoletta.

Nei prossimi numeri sfrutteremo un po' più a fondo il DT1000, nei suoi vari modi di lavoro e utilizzando in modo produttivo il suo tastierino/menu.

La precisione delle misure

Per testare la precisione della tavoletta di MC e del digitizer DT1000 abbiamo realizzato due programmi che prevedono ambedue gli input, oltre ad un input da tastiera, e che permettono di eseguire la misura della superficie di aree e la misura di angoli.

È evidente che quando confrontiamo le tre misure (sicuramente differenti) che otteniamo, rimane irrisolto il problema fondamentale, cioè che non sappiamo quale sia il valore reale; nel nostro caso il valore effettivo dell'area o degli angoli che misuriamo.

Ma questo è un "inconveniente" comune a tutte le misure che si eseguono, e che viene trattato in una specifica materia che si chiama "Teoria degli Errori" e che si può enunciare nel modo che segue:

La determinazione del valore di una gran-

dezza mediante misure è sempre accompagnata da errori, sicché in realtà non è possibile conoscere tale valore ma solo sapere che esso è con grande probabilità contenuto entro certi limiti che saranno tanto più vicini quanto più raffinati sono gli strumenti impiegati ed accurate le misure eseguite.

Per fare un esempio la misura del diametro di una moneta eseguita con un metro da falegname darà valori differenti rispetto a quella ottenuta utilizzando un calibro. Ma anche quest'ultima non è quella reale anche se è probabile che sia vicina a quella reale.

I programmi di test

I programmi di test sono due, anzi uno, anzi due. Il primo utilizza la nota formula di Gauss per il calcolo di un'area di un poligono qualsiasi, e il secondo utilizza la altrettanto nota formula di Carnot per il calcolo degli angoli di un triangolo dati i tre lati.

Il programma è unico in quanto sarebbe stato inutile ripetere tutte le routine di lettura dati che sono comuni ai due test. La parte relativa al calcolo degli angoli viene numerata da 1000 in su e richiama le stesse routine utilizzate dalla parte relativa al calcolo delle aree.

Vediamo ora la prima parte che gira tutta intorno a un main menu che rinvia ai tre modi di input e che visualizza i risultati ottenuti. In figura 7 vediamo l'output ottenuto immettendo, con le tre modalità, i dati relativi all'area F1 della figura 4.

La differenza dei risultati ottenuti con i diversi modi di immettere dati è dell'ordine del 3%, e questo può dipendere da tre fattori di incertezza che potrebbero, nel caso peggiore, sommarsi.

```

90 GOTO 1000
100 REM INIZIALIZZAZIONE
110 DIM SX(3),SY(3),A(3),XX(99),YY(99):SX(1)=1:SY(1)=1
120 L$="-----"
130 REM MAIN MENU
140 TEXT:HOME:VTAB(4):PRINT"CALCOLO DI UN'AREA POLIGONALE
IN MM."
150 PRINT" I DATI VANNO IMMESSI IN MM /10":PRINT L$
160 VTAB(10):PRINT"1 - PUNTI IMMESSI A MANO":TAB(31)A(1)
170 VTAB(12):PRINT"2 - PUNTI IMM VIA M.C. TABLET":TAB(31)A(2)
180 VTAB(14):PRINT"3 - PUNTI IMM VIA MITABLET":TAB(31)A(3)
190 VTAB(16):PRINT"4 - FINE":PRINT:PRINT L$
200 VTAB(21):PRINT"- SCEGLI":INPUT"":SC
210 IF SC < 0 OR SC > 3 THEN HOME:END
220 IF SC = 2 THEN GOSUB 680:GOSUB 740
230 IF SC = 3 THEN GOSUB 810
240 HOME:PRINT"CALCOLO DI UNA AREA POLIGONALE"
250 PRINT"DATI I SUOI VERTICI":PRINT L$:V=1
260 PRINT"VERT. N.":V":TAB(20)
270 IF SC = 1 THEN GOSUB 460:GOTO 300
280 IF SC = 2 THEN GOSUB 480:GOTO 300
290 IF SC = 3 THEN GOSUB 600
300 XX(V)=XX+SX(SC):YY(V)=YY+SY(SC)
310 IF V = 1 THEN XIX=XX(V):YIX=YY(V)
320 REM TEST FINE IMMISSIONE
330 IF SC = 1 AND XX = 999 THEN GOTO 370
340 IF SC = 2 AND XX < X1 AND YY < Y1 THEN GOTO 370
350 IF SC = 3 AND FX = 8 THEN GOTO 370
360 V = V + 1:GOTO 260
370 GOSUB 380:GOSUB 920:GOTO 140
380 REM ROUTINE DI CALCOLO AREA
390 XX(V)=XIX:YY(V)=YIX
400 V = V + 1:XX(V)=XX(2):YY(V)=YY(2)
410 HOME:FOR I = 1 TO V
420 PRINT I:TAB(7)XX(I):TAB(14)YY(I):NEXT I:GOSUB 920
430 S = 0:FOR I = 2 TO V - 1
440 S = S + XX(I) * (YY(I - 1) - YY(I + 1)):NEXT
450 A(SC) = ABS(S / 2) / 100:RETURN
460 REM LETTURA DA TASTIERA
470 INPUT" ":XX:YY:RETURN
480 REM LETTURA X,Y DA DIGITIZER
490 PRINT CHR$(7):
500 GOSUB 550:IF PEEK(49251) > 127 THEN 500
510 A = (Z0 - P0) * PZ:B = (P1 - Z1) * PY - A
520 XX = 150 * (COS(B) - COS(A))
530 YY = 150 * (SIN(B) + SIN(A))
540 PRINT XX,YY:RETURN
550 REM LETTURA CONTINUA VALORI PADDLE
560 POKE 779,100:CALL 768:P0 = FN PK(12)
570 FOR K = 1 TO 50:NEXT K
580 POKE 779,101:CALL 768:P1 = FN PK(12)
590 FOR K = 1 TO 50:NEXT K:RETURN
600 REM LETTURA DA DIGITIZER
610 REM WATANABE MITABLET DT1000
620 PRINT CHR$(7):REM AUTORIZZ. INPUT
630 K$="":IN#2:REM SLOT
640 FOR I = 1 TO 12:GET J$:K$=K$+J$:NEXT I
650 XX = VAL(MID$(K$,1,4)):YY = VAL(MID$(K$,6,4))
660 F% = VAL(MID$(K$,11,1))
670 IN#0:PRINT XX,YY:RETURN
680 REM INIZIALIZZ. MC TABLET
690 PRINT CHR$(4)"LOAD PADDLE CODE"
700 DEF FN PK(I) = PEEK(I) + 256 * PEEK(I + 1)
710 Z0 = FN PK(797):Z1 = FN PK(799)
720 V0 = FN PK(801):V1 = FN PK(803)
730 PY = 3.14159 / V1:PZ = 3.14159 / V0:RETURN
740 REM TARATURA M.C. TABLET
750 HOME:PRINT"TRATTURA TABLET":PRINT
760 PRINT"IMMETTI I PUNTI 1,2,3 DELLO SCHEMA":PRINT L$
770 PRINT:PRINT" PUNTO 1 ":GOSUB 480:X1=XX:Y1=YY
780 PRINT:PRINT" PUNTO 2 ":GOSUB 480:X2=XX:Y2=YY
790 PRINT:PRINT" PUNTO 3 ":GOSUB 480:X3=XX:Y3=YY
800 GOTO 870
810 REM TRATTURA WATANABE MITABLET
820 HOME:PRINT"TRATTURA MITABLET":PRINT
830 PRINT"IMMETTI I PUNTI 1,2,3 DELLO SCHEMA":PRINT L$
840 PRINT:PRINT" PUNTO 1 ":GOSUB 600:X1=XX:Y1=YY
850 PRINT:PRINT" PUNTO 2 ":GOSUB 600:X2=XX:Y2=YY
860 PRINT:PRINT" PUNTO 3 ":GOSUB 600:X3=XX:Y3=YY
870 PRINT:PRINT L$
880 SX(SC)=1000/(X2-X1):SY(SC)=1000/(Y3-Y1)
890 PRINT:PRINT"SCALA SX=":SX(SC):PRINT
900 PRINT:PRINT"SCALA SY=":SY(SC):PRINT
910 PRINT L$:GOSUB 920:RETURN
920 REM LOOP ATTESA
930 FOR H = 1 TO 999:NEXT:RETURN

```

Figura 5 - Listato del programma di test. La scelta delle varie modalità di immissione dati è eseguita tramite l'input di riga 200.

```

JPRINT""
JLIST1000.
1000 REM CALCOLO DI ANGOLI
1010 DIM A(3,4),X(3,5),Y(3,5),L(3,5)
1020 L# = "-----"
1030 TEXT HOME:PRINT "CALCOLO DI ANGOLI"
1040 PRINT "I DATI VANNO IMMESSI IN MM./10":PRINT L#
1050 PRINT "1 - PUNTI IMMESSI A MANO"
1060 PRINT "2 - PUNTI IMM. VIA M. C. TABLET"
1070 PRINT "3 - PUNTI IMM. VIA MITABLET"
1080 PRINT "4 - FINE":PRINT L#
1090 PRINT "ANG. MANO MC TABLET MITABLET":PRINT L#
1100 FOR L = 1 TO 4:IF L = 4 THEN PRINT L#:"TOT.":GOTO 1120
1110 PRINT L#
1120 FOR I = 1 TO 3:PRINT TAB< I + 12 - 6>A(I,L):
1130 NEXT I:PRINT L#
1140 PRINT L#:INPUT " - SCEGLI ":SC
1150 HOME:PRINT "IMMISSIONE DEI PUNTI":PRINT L#
1160 IF SC = 4 THEN END
1170 IF SC = 2 THEN GOSUB 680
1180 FOR L = 1 TO 3:PRINT "PUNTO "L#": "
1190 IF SC = 1 THEN GOSUB 460:GOTO 1220
1200 IF SC = 2 THEN GOSUB 480:GOTO 1220
1210 IF SC = 3 THEN GOSUB 600
1220 X(SC,L) = X:Y(SC,L) = Y: NEXT L
1230 X(SC,4) = X(SC,1):Y(SC,4) = Y(SC,1)
1240 X(SC,5) = X(SC,2):Y(SC,5) = Y(SC,2)
1250 REM CALCOLO ANGOLI
1260 REM TRAMITE IL TEOREMA DI CARNOT
1270 FOR L = 1 TO 4
1280 X1 = X(SC,L):X2 = X(SC,L + 1):DX = (X2 - X1) ^ 2
1290 Y1 = Y(SC,L):Y2 = Y(SC,L + 1):DY = (Y2 - Y1) ^ 2
1300 L(SC,L) = SQR (DX + DY)
1310 NEXT L:L(SC,5) = L(SC,2):A(SC,4) = 0
1320 FOR L = 2 TO 4
1330 C(SC,L) = (L(SC,L - 1) ^ 2 + L(SC,L + 1) ^ 2 - L(SC,L) ^ 2) / (
2 + L(SC,L - 1) + L(SC,L + 1))
1340 CS = C(SC,L)
1350 A(SC,L + 1) = - ATN (CS / SQR (CS + CS + 1)) + 1.5708
1360 A(SC,L - 1) = A(SC,L - 1) + 180 / 3.1416
1370 A(SC,4) = A(SC,4) + A(SC,L - 1)
1380 NEXT L:GOTO 1020
    
```

Figura 7 - Listato del programma di test degli angoli. Il programma conserva tutti i dati relativi alle modalità di input dei dati.

CALCOLO DI UN'AREA POLIGONALE IN MMQ.
I DATI VANNO IMMESSI IN MM./10

1 - PUNTI IMMESSI A MANO	3619.625
2 - PUNTI IMM. VIA M. C. TABLET	3505.05
3 - PUNTI IMM. VIA MITABLET	3583.7
4 - FINE	

- SCEGLI

Figura 6 - Output sul video del programma di calcolo dell'area. La misura dell'area si può eseguire anche più volte con la stessa modalità.

CALCOLO DI ANGOLI
I DATI VANNO IMMESSI IN MM./10

1 - PUNTI IMMESSI A MANO			
2 - PUNTI IMM. VIA M. C. TABLET			
3 - PUNTI IMM. VIA MITABLET			
4 - FINE			

ANG.	MANO	MC. TABLET	MITABLET
1	58.8160124	60.0285752	60.2611164
2	59.4774253	58.4710857	60.1514308
3	61.7067727	61.5005496	59.5876633
TOT	180.00021	180.00021	180.00021

- SCEGLI

Figura 8 - Output sul video del programma di test degli angoli. Il triangolo di figura 4 risulta essere equilatero a meno di un errore inferiore al grado.

— Precisione del disegno. Nel nostro caso il disegno è realizzato con un pennino 0.8 che quindi ha un suo spessore e una propria superficie in funzione della scala e delle dimensioni della figura.

— Precisione di puntamento. Dipende dall'operatore che può o meno puntare correttamente il disegno, in funzione sia dello strumento di puntamento che della fermezza della mano.

— Precisione dello strumento. Precisione che è inutile che sia di ordine di grandezza superiore alle prime due.

Anche nel caso della misura di angoli l'ordine di grandezza dell'errore è del 3 - 4% per gli stessi motivi citati in precedenza. Prove realizzate con figure più grandi hanno diminuito tali percentuali.

È infatti evidente che l'errore si può valutare o come errore assoluto (misurato con la stessa unità di misura della grandezza oggetto della misura) es. 1 mm, o come errore relativo misurato con numeri adimensionali, cioè misura assoluta dell'errore su misura assoluta della grandezza. Per esempio 1 mm su 10 cm rappresenta un errore dell'1 per cento.

Con l'aumentare della dimensione della grandezza e a parità di errore assoluto diminuisce l'errore relativo. In pratica se dobbiamo misurare l'area di una superficie, la misura sarà tanto più precisa quanto più grande sarà la superficie stessa.

Essendo questo un discorso sugli errori di misura noioso e in fondo ovvio, passiamo ad esaminare i programmi.

Il primo, listato in figura 5, comprende tutte le routine che servono anche al secondo:

- lettura da tastiera 460-470
- lettura da MC tablet 480-590

— lettura da Mitablet 600-670

— inizializzazione MC tablet 690-730

queste routine sono già state viste nei programmi specifici presentati prima.

Vi sono poi due routine di taratura della MC tablet e del Digitizer Watanabe.

I dati immessi da tastiera vanno immessi in decimi di millimetro. Utilizzando le periferiche occorre operare uno scaling.

I due segmenti ortogonali sono lunghi 10 cm e puntandoli, nella routine di taratura, vengono settate le scale con le quali vanno interpretate le coordinate immesse.

Operata la scelta, e quindi definito il valore SC di riga 200, vengono svolte le varie routine alternative, vengono caricati i due vettori X%, Y% con i dati già sottoposti a scaling.

La fine immissione (routine di righe 320 e 350) è ottenuta in tre modi differenti. Se si lavora da tastiera occorre immettere il valore X% = 999.

Se si lavora da MC tablet occorre puntare un punto in alto a sinistra rispetto al punto 1, immesso con l'operazione di taratura. Se infine si lavora con la Mitablet occorre premere il tasto 3 del cursore.

Finita l'immissione viene richiamata la routine da calcolo area che, come detto, usa la formula di Gauss:

$$A = \frac{\sum (x_i) \cdot (y_{i-1} - y_{i+1})}{2}$$

Per lavorare con tale formula occorre allungare di due unità la dimensione dei vettori con i dati (righe 390-400).

Eseguito il calcolo dell'area, viene riproposto il menu, per un nuovo tipo di calcolo o per la fine.

Il programma di test per gli angoli

La formula di Carnot per il calcolo di un angolo di un triangolo dati i suoi tre lati è:

$$A^2 = B^2 + C^2 - 2 \cdot B \cdot C \cdot \cos(\alpha) \text{ cioè } \alpha = \arccos\left(\frac{B^2 + C^2 - A^2}{2 \cdot B \cdot C}\right)$$

Quindi dati tre punti A,B,C sul piano si può calcolare l'angolo formato dalle coppie di segmenti uscenti da ciascuno dei punti e congiungenti i rimanenti due, il che è assolutamente equivalente a dare i tre vertici di un triangolo per cercarne i tre angoli.

Il programma è listato in figura 7 e produce sul video l'output di figura 8.

Come detto la numerazione parte da 1000 ma richiede le routine di righe 460, 480, 600, 680 del programma precedente. Ese i due programmi si uniscono occorrerà digitare RUN 1000 per eseguire la parte di calcolo degli angoli.

Righe 1000-1020 inizializzazione

Righe 1030-1080 main menu

Righe 1090-1140 tabella dei risultati, che appare con i menu, in quanto si può eseguire la routine di input più volte e vedere via via i risultati.

Righe 1150-1170 inizio delle tre routine di input. Ricordiamo che se si usa la MC tablet va eseguita la routine 680 per la sua inizializzazione.

Righe 1180-1220 routine di input vera e propria.

I dati vengono caricati nelle matrici X(3,5), Y(3,5) dove il primo indice indica la scelta di modalità di input operata, il secondo indica l'ordine di immissione dei punti. I punti sono 5 e non 3, in quanto il 4

e il 5 vengono posti uguali al punto 1 e 2 (righe 1230-1240), questo, come al solito, per poter eseguire i calcoli utilizzando dei loop.

Dalla riga 1270 in poi c'è la routine che esegue i calcoli secondo il teorema di Carnot. Dapprima vengono calcolati i lati del triangolo, che vengono immagazzinati nella matrice L(3,5), poi (riga 1330) viene calcolata la formula di Carnot che dà il $\text{COS}(\alpha)$. Poiché non esiste nel BASIC standard la funzione trigonometrica ARCOS, la si ricava tramite la ATN (riga 1350).

Infine l'angolo trovato viene tradotto da radianti in gradi. Eseguendo il loop per il calcolo degli angoli, viene calcolata anche la somma e il tutto viene immagazzinato nella matrice A(3,4).

Conclusioni

Per concludere. La precisione è un concetto che deve essere ben chiaro a chi utilizza un computer. Una maggior precisione di calcolo si paga sempre in termini di occupazione di memoria e di velocità di elaborazione. Va dunque utilizzata la precisione più adatta al problema che si sta trattando.


Questo discorso vale anche in computer grafica, specie quando si utilizzano periferiche per l'input e per l'output dei dati.

Infatti se si utilizza, come output, il video grafico del computer, il calcolo e l'input possono essere adeguatamente precisi. Ma se l'output è un plotter oppure l'applicazione richiede risultati matematicamente precisi si deve adeguare sia la fase di

input sia la fase di elaborazione.

In particolare, se i nostri programmi hanno come output il video grafico dell'Apple II, la tavoletta grafica di MCmicrocomputer, che è dotata di un potente SW di base, è una periferica di input di adeguate prestazioni.

Ma se i nostri obiettivi sono di computer grafica produttiva, tutto il sistema va potenziato e quindi occorrono sia un digitizer che un plotter di adeguate prestazioni.

Avendo a disposizione sia il Plotter MP1000 che il digitizer DT1000 della Watanabe, che formano un "coordinato" di buone prestazioni (a basso costo), approfondiremo nei prossimi numeri il tema della precisione, che è l'indice principale delle prestazioni di un apparecchio per la computer grafica. 

Digitizer Watanabe DT1000 Mitablot

Come si intuisce dalla sigla, il digitizer Watanabe DT1000 appartiene alla stessa famiglia del plotter MP1000, recentemente provato su MCmicrocomputer, del quale ha le stesse prestazioni in termini di precisione e area di lavoro, e con il quale quindi costituisce una "linea coordinata" per la computer grafica.

Ha una area di lavoro di 380 per 260 millimetri con una risoluzione di 0.1 mm. Questo vuol dire che il digitizer invia al computer coordinate in decimi di millimetro. L'accuratezza ovvero la precisione di puntamento, che è del tutto indipendente dai valori trasmessi, è invece di 1 mm se si lavora sul formato massimo.


Il DT1000 ha in dotazione un ottimo cursore con traguardo ottico dotato di una lente di ingrandimento e che permette quindi un puntamento più preciso possibile, anche se qui, intervenendo l'uomo, la precisione dipende dalla sua "mira". Tra gli accessori disponibili c'è anche un puntatore a stilo.

Un'altra caratteristica del DT1000 è il tastierino, a quattro pulsanti, solidale con il cursore. In pratica il digitizer invia al computer tre dati $X\%$, $Y\%$, $F\%$. Il simbolo di percento indica che il dato è in forma intera. $X\%$, $Y\%$ sono le coordinate del punto e $F\%$ è il segnale che indica quale dei quattro tasti è stato premuto.

Questo permette in programmi, non molto complessi, che utilizzano il digitizer per l'input, di evitare la costruzione del menu, e quindi di sfruttare per i dati grafici tutta l'area, e inoltre di semplificare molto il pro-



Particolare del cursore della Mitablot. Il tastierino ha quattro tasti 1,2,3, Z che producono i valori 1,2,4 e 8.



Il digitizer Watanabe Mitablot DT1000. La superficie è inclinata di qualche grado, la dimensione dell'area riservata all'input, è di 380 per 260 millimetri.



Particolare del lato posteriore della Mitablot. Qui sono alloggiati connessioni, switch, tasti, ecc.

gramma che non deve avere routine di riconoscimento area menu e, all'interno di questa, di riconoscimento delle singole aree delle opzioni.

Per quanto riguarda l'interfacciamento, la macchina è dotata della classica RS 232 che lavora in input/output in quanto molti dei suoi comandi possono venire via software.

Esaminando il lato posteriore si notano

dalla sinistra: la connessione dello stilo o cursore, la connessione 232 a 25 poli, 2 switch per il settaggio delle caratteristiche di comunicazione, una connessione per il display a led, il tasto reset, terra, fusibile, alimentazione e interruttore ON/OFF.

Nei prossimi numeri proveremo a fondo la macchina ed in queste occasioni descriveremo i vari comandi.

al servizio
dei tuoi problemi...



...easy·byte ti suggerisce
il nome, le periferiche, il software
ed il prezzo.

RIVENDITORE AUTORIZZATO:

 apple computer

 *Lisa*

MacIntosh

olivetti M20

VICTOR

sinclair  commodore

SOCIO:



risorse, idee e soluzioni

 **COMPUTER SHOP**
easy·byte s.r.l.


ROMA:
Via G. Villani, 24-26
Tel. (06) 7811519-7887926

Sede operativa, centro ricerca
e sviluppo software
Via B. Platina, 22
Tel. (06) 786246

LATINA:
Via Enrico Toti (Galleria CISA)
Tel. (0773) 488001


"PIXY 3" DISEGNA A 3 MANI PER IL TUO PERSONAL COMPUTER.

advertteam



"PIXY 3" è un plotter professionale, ma a prezzo contenuto, che unisce alla massima affidabilità prestazioni elevate.

Formato A4, tre colori, velocità di scrittura di 200 mm/sec, compatibile con i packages software più noti è particolarmente studiato per soddisfare le esigenze grafiche dei Personal Computers. Riproduce informazioni grafiche, diagrammi e statistiche sia su carta comune che su lucidi utilizzabili per proiezione.



 **MANNESMANN
TALLY**

20094 Corsico (MI) - Via Cadamosto, 3
Tel. (02) 4502850/855/860/865/870
Telex 4500934
00137 Roma - Via I. Del Lungo, 42
Tel. (06) 8278458
10099 San Mauro (TO) - Via Casale, 308
Tel. (011) 8225171
40050 Monteveglio (BO) - Via Einstein, 5
Tel. (051) 965208